

LA-10975

UC-10

Issued: June 1987

LA--10975

DE87 010834

An Improved, Computer-Based, On-Line Gamma Monitor for Plutonium Anion Exchange Process Control

Noah G. Pope
S. Fredric Marsh

DISCLAIMER

This report was prepared as an account of work sponsored by an agency of the United States Government. Neither the United States Government nor any agency thereof, nor any of their employees, makes any warranty, express or implied, or assumes any legal liability or responsibility for the accuracy, completeness, or usefulness of any information, apparatus, product, or process disclosed, or represents that its use would not infringe privately owned rights. Reference herein to any specific commercial product, process, or service by trade name, trademark, manufacturer, or otherwise does not necessarily constitute or imply its endorsement, recommendation, or favoring by the United States Government or any agency thereof. The views and opinions of authors expressed herein do not necessarily state or reflect those of the United States Government or any agency thereof.

MASTER

Los Alamos Los Alamos National Laboratory
Los Alamos, New Mexico 87545

DISTRIBUTION OF THIS DOCUMENT IS UNLIMITED

EAB

AN IMPROVED, COMPUTER-BASED, ON-LINE GAMMA MONITOR FOR PLUTONIUM ANION EXCHANGE PROCESS CONTROL

by

Noah G. Pope and S. Fredric Marsh

ABSTRACT

An improved, low-cost, computer-based system has replaced a previously developed on-line gamma monitor. Both instruments continuously profile uranium, plutonium, and americium in the nitrate anion exchange process used to recover and purify plutonium at the Los Alamos Plutonium Facility. The latest system incorporates a personal computer that provides full-feature multichannel analyzer (MCA) capabilities by means of a single-slot, plug-in integrated circuit board. In addition to controlling all MCA functions, the computer program continuously corrects for gain shift and performs all other data processing functions. This Plutonium Recovery Operations Gamma Ray Energy Spectrometer System (PROGRESS) provides on-line process operational data essential for efficient operation. By identifying abnormal conditions in real time, it allows operators to take corrective actions promptly. The decision-making capability of the computer will be of increasing value as we implement automated process-control functions in the future.

INTRODUCTION

The major aqueous process used to recover and purify plutonium at the Los Alamos Plutonium Facility is anion exchange in nitric acid. This process is nearly ideal for separating plutonium from a wide variety of impure materials, as the anionic nitrate complex of Pu(IV) is more strongly sorbed than any other complex ion, and few other elements show even moderate sorption from nitric acid.¹

Operators at Los Alamos have traditionally followed a fixed procedure, even though feed solutions vary over a broad range of impurity compositions and concentrations. Before the earlier on-line gamma monitor² became available less than two years ago,

process operators based most process-control decisions on only their visual observations and intuition.

Although the earlier version of the on-line gamma monitor significantly improved the efficiency of this anion exchange process, it lacked the reserve computational power and flexibility needed to support a parallel Los Alamos development effort directed toward the automation of major aqueous processes.

The objective of this work was to develop a compact, low-cost, computer-based version of the on-line gamma monitor. This improved version would provide sufficient reserve capability to compile real-time process-control data from a variety of monitors during future automated process-control applications.

INSTRUMENTATION AND EQUIPMENT

Gamma Detector. High-purity, coaxial germanium detector of ≈ 1.80 -keV resolution and $\approx 15\%$ efficiency, in a horizontal integral configuration. Canberra Industries, Meriden, Connecticut.

Bias Supply. Single-width NIM bias supply, low noise, 0 to ± 5 kV.

Computer. DataCAT, IBM-compatible personal computer, 512-K RAM, with dual floppy-disk drives. Datacraft, Inc., Gardena, California. Includes HARD-CARD 10-megabyte fixed disk drive. Plus Development Corp., Milapitas, California.

Multichannel Analyzer. Model PCA-2000 single-slot, plug-in board with 2048-channel ADC and memory. Includes single-channel analyzer, multichannel scaler card, and software for acquisition and display. The Nucleus, Inc., Oak Ridge, Tennessee.

Spectroscopy Amplifier. Single-width NIM spectroscopy amplifier and pile-up rejecter, model 572. EG&G ORTEC, Oak Ridge, Tennessee.

NIM Bin. Model 2100 bin/power supply, 150 W. Canberra Industries, Meriden, Connecticut.

Plotter. Model 695 PC plotter, four-pen, with RS-232-C serial interface. Houston Instruments, Austin, Texas.

Radioisotope Calibration Source. Barium-133, 0.1- μ Ci, sealed solid source, taped to detector to provide reference gamma energies for gain stabilization. Isotope Products Laboratories, Burbank, California.

Disk Operating System. DOS Version 2.0 or higher. IBM Corporation, Boca Raton, Florida.

Compiler. IBM Basic Compiler, Version 2.01, 1985. IBM Corporation, Boca Raton, Florida.

SYSTEM INSTALLATION AND OPERATION

Although ^{239}Pu and ^{241}Am have gamma rays that are directly suitable for passive gamma assay, the natural and depleted uranium impurities in typical Los Alamos feed materials do not. A novel and indirect

radiotracer technique therefore is used. The position of any uranium impurity that passes through the anion exchange process is traced by ^{237}U , a minor alpha-decay daughter of ^{241}Pu that is always present in plutonium processed at Los Alamos. This radiotracer technique is detailed in a separate report.²

A high-purity germanium gamma detector monitors the outlet stream from the anion exchange column (Fig. 1) as it flows through an existing three-quarter-inch stainless steel process pipe. The detector is shielded from background radiation by a lead shield (minimum 1.5-inch thickness) that contains a lining of 0.030-inch cadmium to absorb fluorescent x rays from lead. Another cadmium absorber is inserted between the pipe and the detector to attenuate a major portion of the abundant 59.5-keV gamma rays from ^{241}Am (Fig. 2).

The complete on-line gamma monitor system (Fig. 3) provides real-time elution profiles of americium, uranium, and plutonium (Fig. 4) as these elements elute from the anion exchange column. The changing detector signals reflect the varying composition of the outlet stream from the anion exchange column as (1) americium impurity elutes first, (2) uranium and other impurities elute next, and finally (3) purified plutonium elutes after being reduced from strongly sorbed Pu(IV) to nonsorbed Pu(III). These recorded elution profiles provide data from which operators can optimize many process parameters. The stripchart also provides a permanent record of the daily performance for each anion exchange system.

COMPUTER PROGRAM SYNOPSIS

The Plutonium Recovery Operations Gamma Ray Energy Spectrometer System (PROGRESS) program that we developed consists of nearly 1000 lines of BASIC code that controls all data acquisition, processing, and output operations. Software for the PCA multichannel analyzer plug-in card (Fig. 5), supplied by The Nucleus, Inc., must be loaded manually before the BASIC program can be executed. The PROGRESS program (see Fig. 6) then controls the vendor-supplied software. (Appendix A is a detailed description of this program.)

Many parameters and functions of the PCA-2000 plug-in card can be directly accessed from a high-level interface language. BASIC was selected for this application because it is easily programmed and is capable of interfacing with assembly-language subroutines by




Fig. 1. High-purity germanium detector monitoring the outlet pipe from an anion exchange system at Los Alamos.




Fig. 2. Details of shielding and mounting configuration of the gamma detector.




Fig. 3. Complete on-line gamma monitor system installed under glove box.




Fig. 4. Typical stripchart record from the anion exchange separation process.




Fig. 5. The PCA-2000 single-slot, plug-in integrated circuit board that provides complete multichannel analyzer capability to a personal computer. (Only a single board is required; this photograph shows both sides of the single board.)




Fig. 6. Flow chart for major sections of the PROGRESS program.

means of POKE statements and PEEK and USR functions. These commands allow subroutines, flags, and pointers in the PCA software to be read directly into or out of memory locations on the PCA board. The detailed information required to directly access PCA software and firmware was obtained from The Nucleus, Inc.^{3,4}

Microsoft Corporation's GW-BASIC was selected for compatibility with the DataCAT computer. (The DataCAT computer was selected because its construction and small size make it much better suited for a chemical plant environment than are competitive computers designed for office use.)

Vendor-supplied subroutines directly (1) change the cursor position, (2) update and display the spectrum, (3) calculate the region-of-interest (ROI) centroids and net integrals, (4) preset data acquisition times, and (5) start and stop data acquisition. Direct alteration of memory locations that contain logical on-and-off byte configurations accomplishes certain other operations, such as setting and clearing ROIs.

Many functions require that parameters or data be converted from regular decimal notation to multiple-byte binary or packed binary coded decimal (BCD) format. Large portions of several subroutines therefore consist of algorithms that accomplish these conversions.

PROGRAM STRATEGY

High-purity germanium gamma detectors routinely produce high-resolution gamma peaks. Gamma spectral measurements of high-resolution peaks are most reliable when the ROI is only slightly wider than the peak being measured. Such tightly defined ROIs, however, leave the measurement system susceptible to large errors if peak positions shift during the measurement period. An external electronic module typically is used to prevent gain-shift errors. Such modules automatically sense gain shift in the analog-to-digital converter (ADC) and adjust as required to correct for the detected shift. Unfortunately, the internal ADC circuitry on a PCA-2000 plug-in board is inaccessible to such external electronic gain stabilizers.

Instead, PROGRESS uses two gamma rays (one at 81 keV and the other at 356 keV) from a ¹³³Ba calibration source as gamma-energy reference points

for a unique software-controlled gain stabilization. PROGRESS first locates these two peaks, which nearly bracket the gamma-energy range of interest, and then determines the centroid for each. Based on the channel positions of these two calibration peaks, ROI positions are calculated and reset for ²⁴¹Am (59.5 keV), ²³⁹Pu (129 keV), and ²³⁷U (208 keV). This recalibration sequence is automatically repeated approximately every 35 seconds.

PROGRESS then initiates a spectrum acquisition loop (preset for 30 seconds in our application). Updated, cumulative spectral data are displayed on the video screen approximately every second during the acquisition period. At the end of the acquisition period, the program interrogates each preset ROI and calculates the net integral for each peak. These values are converted to a predetermined logarithmic scale and are output to the plotter during the subsequent data acquisition cycle.

The PROGRESS program may be executed in either an interpretive or compiled mode. The version compiled by an IBM BASIC Compiler requires approximately 35 seconds to execute a complete cycle (of which 30 seconds is data acquisition time). This is approximately 5 seconds less than the 40 seconds required to execute the interpretive version. Another compiler, IBM QuickBASIC, was rejected because it was incompatible with several essential BASIC commands.

The complete PROGRESS program (see Appendix B) includes numerous comment statements in the code. Although these statements consume a small amount of computer time in the interpretive mode, they consume no time in the compiled mode and they serve as a diagnostic aid to users who are unfamiliar with the software.

ACKNOWLEDGMENTS

The authors are grateful to Dan Blankenship of The Nucleus, Inc., for consultation concerning firmware access and programming strategy, to James Dyke for his helpful discussions and programming suggestions, and to Randy Vaughn for assistance during the installation and demonstration of the on-line monitor on a full-scale plutonium process system.

REFERENCES

1. J. P. Faris and R. F. Buchanan, "Applications of Anion-Exchange Spectrographic Procedures in Nitric Acid Medium," U. S. Atomic Energy Commission report TID-7606, pp. 185-194 (1960).
2. S. F. Marsh and M. C. Miller, "Plutonium Process Control Using an Advanced On-Line Gamma Monitor for Uranium, Plutonium, and Americium," Los Alamos National Laboratory report LA-10921 (May 1987).
3. Nucleus Personal Computer Analyzer Operating Instructions, August 1986 version, The Nucleus, Inc., Oak Ridge, Tennessee.
4. Dan Blankenship, The Nucleus, Inc., Oak Ridge, Tennessee, personal communications (1986).

APPENDIX A. DETAILED DESCRIPTION OF PROGRAM

The PROGRESS program consists of six major sections, each of which involves supporting subroutines, as shown in Fig. 6.

MAJOR PROGRAM SECTIONS

User-Defined Parameters

Detailed below are certain essential parameters that initially must be defined by the user. These parameter definitions are retained as default settings unless they are intentionally changed.

Barium-133 Calibration Source. Gamma peaks at 81 keV and 356 keV from the ^{133}Ba radioisotope calibration source are used to recalibrate the system after each cycle. The program prompts the operator to define the channel numbers that bracket each of these two ROIs. These ROI settings are entered manually in the PCA-2000 mode before program execution.

Data Acquisition Time. The data acquisition time must be specified by the operator; a time of 30 seconds typically is used.

Plotter Scale. The y-axis of the plotter represents the log of the net peak area, which typically is divided into four decades that cover the range of 1 to 10,000. The scale assigned to the x-axis (time) is specified by the operator to allow the output to be expanded or contracted according to the estimated time required for a given process run. This time-scale control is achieved by varying the center-to-center spacing between adjacent plotted characters in increments of 0.001 inch. Default values are defined separately.

System Initialization

After the operator initializes the system using parameters selected above, the PROGRESS program controls the following operations: (1) The blank plotter paper is scaled and advanced to the proper starting position. (2) The MCA display screen is activated to update and display the spectral data and system parameters. (3) The data acquisition

time (Preset Time Function) is converted from seconds to a 3-byte binary coded decimal and executed. (4) The ROI function is set to the "ON" position.

Energy Calibration

ROIs for the two gamma peaks of the ^{133}Ba radioisotope calibration source, as defined in the PARAMETER INPUT section, are set and their centroid locations are calculated. The centroids are assigned gamma energy values of 80.998 keV and 356.005 keV, respectively. The linear relationship between gamma energy and MCA location is calculated as $C = mE + b$, where C is channel number and E is energy. The slope value, m , is equal to the number of channels that separate the centroids of the 81-keV and 356-keV peaks divided by the energy that separates them. Substitution of the channel and energy values for either peak into the equation will provide the value of b , the zero intercept.

From this equation the peak centroid positions are calculated for gamma energies of 59.5 keV, 129 keV, and 208 keV, used to monitor ^{241}Am , ^{239}Pu , and ^{237}U , respectively (Fig. 4). The number of channels to the left and the number of channels to the right of the peak centroid that defines the ROI for each peak are preset, based on empirical peak resolution values, which are a function of the peak energy and system resolution.

Spectral Acquisition

Previously collected data are cleared by the Clear Spectral Data function. A new acquisition cycle then is initiated by setting the START/STOP Data Acquisition function to the "ON" position. During the new data acquisition cycle, processed data from the previous acquisition and calculation cycle are output to the plotter. A print buffer in the Houston Instruments plotter allows these data to be quickly transferred without interference to the ongoing acquisition cycle.

Two alternate function-key combinations may be used during the spectral acquisition loop. The Alt and F1 keys may be simultaneously pressed to

record the position of significant events with a special mark at the top of the plotter paper. Simultaneous depression of the Alt and F10 keys terminates the data acquisition loop and redisplay an options menu. From this options menu the operator may elect to (1) resume the measurement cycle in progress, (2) begin a new measurement cycle, or (3) exit the program and return to DOS.

If neither alternate function-key option is invoked, data acquisition continues until the flag for data acquisition preset time registers "OFF" or timed-out status.

ROI Net Integral Calculations

The net integral of each ROI is sequentially calculated by invoking a Net Integral Calculation function within the subroutine for reading net integrals. This algorithm is repeated for each of the specified ROIs. The ROI being processed is designated by placing the MCA display cursor within that ROI.

Clear ROIs

The ROIs used to monitor ^{241}Am , ^{239}Pu , and ^{237}U are cleared after each acquisition cycle is complete, in preparation for a new energy calibration. Program control then returns to the Energy Calibration algorithm at the appropriate portion of the program (Fig. 6).

SUPPORTING SUBROUTINES

Set and Clear ROIs

The vendor-supplied Set and Clear ROI function requires that each channel included in an ROI be addressed and interrogated on an individual basis. This time-consuming procedure requires approximately 0.3 second per channel, or 3 seconds for a 10-channel ROI. The more efficient subroutine developed in the PROGRESS program allows the ROI to be set and cleared by manipulating a portion of memory that contains an ROI "ON/OFF" flag. This subroutine can set or clear a 15-channel ROI in less than 1 second.

PEEK-ing

The BASIC language PEEK function, used in conjunction with the POKE statement, allows values stored in memory locations to be read directly. This subroutine allows the program to interrogate important memory flags and pointers, such as the preset time or data acquisition flag and the RC status flag.

Using Poke Statement to Replace Value in Memory

It can be advantageous to "trick" the PCA-2000 by replacing existing values in memory with new, more useful values. For example, the vendor-supplied function that moves the display cursor to another location must be invoked repeatedly until the desired channel location is reached. Alternatively, the value in the memory location that contains the current cursor position may be replaced with the value of the desired new cursor position. When the PCA-2000 checks for the current cursor location, it finds this new value and responds by moving the cursor to the desired position in a single operation. This subroutine can save 5 to 10 seconds of processing time per cycle.

POKE-ing

Values may be placed directly in selected memory locations using the BASIC language POKE statement. Several vendor-supplied functions are invoked by POKE-ing a function number into the appropriate memory location. These functions include Data Acquisition Start/Stop, Spectral Data Clearing, Single-Channel Cursor Movements, Centroid Calculations, and Net Integral Calculations.

Centroid Calculation

The vendor-supplied Centroid Calculation function requires that the cursor be positioned within the selected ROI. This subroutine directs the program to another subroutine that moves the cursor, as well as invokes the centroid calculation function. The calculated centroid value then is converted from packed binary coded decimal into regular decimal notation for use in subsequent calculations.

Net Integral Calculation

The vendor-supplied Integral Calculation function is invoked and read (PEEK-ed) out of memory by this subroutine. The value then is converted from a binary number into decimal notation for subsequent manipulation.

APPENDIX B. PROGRESS: THE COMPLETE PROGRAM

```
10 KEY OFF
20 SCREEN 0,0,0
30 DEF SEG
40 REM
50 REM      ON-LINE GAMMA MONITOR FOR
60 REM      ANION EXCHANGE SEPARATION PROCESS
70 REM
80 REM      using the PERSONAL COMPUTER ANALYZER (PCA-2000) SYSTEM
90 REM      from THE NUCLEUS, Inc. Oak Ridge, Tennessee
100 REM
110 REM      programmed by
120 REM
130 REM      NOAH G. POPE
140 REM
150 REM
160 REM      NUCLEAR MATERIALS PROCESS TECHNOLOGY GROUP (MST-12)
170 REM      RESEARCH, DEVELOPMENT, AND DEMONSTRATION SECTION
180 REM      LOS ALAMOS NATIONAL LABORATORY
190 REM      LOS ALAMOS, NM 87545
200 REM
210 REM
220 REM      MARCH, 1987
230 REM
240 REM
250 REM*****
260 REM*****
270 REM
280 REM
290 REM*****
300 REM***** SET UP WELCOME SCREEN *****
310 REM*****
320 REM
330 CLS
340 COLOR 13:LOCATE 3,23:PRINT"P":COLOR 11:LOCATE 3,24:PRINT" l u t o n i u m"
350 COLOR 13:LOCATE 5,23:PRINT"R":COLOR 11:LOCATE 5,24:PRINT" e c o v e r y"
360 COLOR 13:LOCATE 7,23:PRINT"O":COLOR 11:LOCATE 7,24:PRINT" p e r a t i o n s"
370 COLOR 13:LOCATE 9,23:PRINT"G":COLOR 11:LOCATE 9,24:PRINT" a m m a"
380 COLOR 13:LOCATE 11,23:PRINT"R":COLOR 11:LOCATE 11,24:PRINT" a y"
390 COLOR 13:LOCATE 13,23:PRINT"E":COLOR 11:LOCATE 13,24:PRINT" n e r g y"
400 COLOR 13:LOCATE 15,23:PRINT"S":COLOR 11:LOCATE 15,24:PRINT" p e c t r o s c o p y"
410 COLOR 13:LOCATE 17,23:PRINT"S":COLOR 11:LOCATE 17,24:PRINT" y s t e m"
420 COLOR 14:LOCATE 19,23:PRINT"Nuclear Materials Process Technology Group MST-12"
430 LOCATE 20,23:PRINT"Research, Development and Demonstration"
440 COLOR 18:LOCATE 22,23:PRINT"ENTER ACCESS CODE:"
450 LOCATE 22,40:COLOR 16:INPUT ACCESS$
460 REM
470 REM||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||
480 REM
490 REM      ACCESS IS A STRING VARIABLE THAT CAN BE CHANGED IN THIS SECTION.
500 REM
510 REM||||||||||||||||||||||||||||||||||||||||||||||||||||||||||||
```

PROGRESS continued

```
520 REM
530 REM
540 IF ACCESS$="GAMMA" THEN 580
550 BEEP:BEEP
560 LOCATE 23,23:COLOR 10:PRINT"INCORRECT- RE-ENTER CODE"
570 GOTO 440
580 REM
590 COLOR 11
600 LOCATE 23,23:PRINT"
610 REM
620 REM
630 REM
640 REM      BRANCH TO THE PARAMETER SETTING LOOP OR TO DEFAULT
650 REM      COLUMN MONITOR (MAIN PROGRAM BODY).
660 REM
670 REM
680 REM
690 LOCATE 22,23:INPUT"ENTER OPTION: (1) SET PARAMETERS or (2) MONITOR
COLUMN:",CHOOSE.OPTION$
700 REM
710 IF CHOOSE.OPTION$="1" THEN 780
720 IF CHOOSE.OPTION$="2" THEN 1110
730 BEEP:BEEP
740 LOCATE 23,23:COLOR 10:PRINT"INCORRECT- RE-ENTER CODE"
750 GOTO 690
760 END
770 REM
780 REM*****
790 REM***** PARAMETER SETTING LOOP *****
800 REM*****
810 CLS
820 COLOR 12
830 LOCATE 3,23:PRINT"PROGRESS Parameters"
840 COLOR 14
850 LOCATE 5,23:PRINT"Input values at the following prompts."
860 LOCATE 6,23:PRINT"(Default values are in parentheses.)"
870 COLOR 14:LOCATE 22,23:PRINT"Nuclear Materials Process Technology Group (MST-12)"
880 LOCATE 23,23:PRINT"Research, Development and Demonstration"
890 COLOR 10
900 LOCATE 8,23:INPUT"PLOTTER PAPER ADVANCE INTERVAL (6)=";ADVANCE.INTERVAL
910 IF ADVANCE.INTERVAL=0 THEN ADVANCE.INTERVAL=6
920 PAPER.ADVANCE=12
930 COLOR 11
940 LOCATE 10,23:INPUT"PRESET TIME IN SECONDS (30)=";PRESET.TIME
950 IF PRESET.TIME=0 THEN PRESET.TIME=30
960 COLOR 12
970 LOCATE 12,23:INPUT"1ST Ba ROI LEFT SIDE in channels (338)=";BA.1.ROISTART
980 IF BA.1.ROISTART=0 THEN BA.1.ROISTART=338
990 LOCATE 13,23:INPUT"1ST Ba ROI RIGHT SIDE in channels (356)=";BA.1.ROIEND
1000 IF BA.1.ROIEND=0 THEN BA.1.ROIEND=356
1010 COLOR 13
1020 LOCATE 15,23:INPUT"2ND Ba ROI LEFT SIDE in channels (1538)=";BA.2.ROISTART
1030 IF BA.2.ROISTART=0 THEN BA.2.ROISTART=1538
1040 LOCATE 16,23:INPUT"2ND Ba ROI RIGHT SIDE in channels (1557)=";BA.2.ROIEND
```

PROGRESS continued

```
1050 IF BA.2.ROIEND=0 THEN BA.2.ROIEND=1557
1060 COLOR 14
1070 LOCATE 18,23:INPUT"DO YOU WANT TO MAKE A CORRECTION?";CORRECTIONS$
1080 IF CORRECTIONS$="Y" OR CORRECTIONS$="y" THEN I70
1090 REM
1100 GOTO 1310
1110 REM*****
1120 REM***** DEFAULT PARAMETERS *****
1130 REM*****
1140 REM
1150 REM THESE DEFAULT PARAMETERS ARE FOR A HOUSTON INSTRUMENTS PLOTTER
1160 REM AND FOR A SPECIFIC DETECTOR.
1170 REM
1180 ADVANCE.INTERVAL=6 ' DETERMINES THE SPACING BETWEEN DATA POINTS
1190 PAPER.ADVANCE=12 ' DETERMINES THE PLOTTER STARTING POINT
1200 PRESET.TIME=30 ' PRESET TIME (IN SECONDS)
1210 BA.1.ROISTART=338 ' LEFT SIDE OF THE 81-KeV ROI
1220 BA.1.ROIEND=356 ' RIGHT SIDE OF THE 81-KeV ROI
1230 BA.2.ROISTART=1538 ' LEFT SIDE OF THE 356-KeV ROI
1240 BA.2.ROIEND=1557 ' RIGHT SIDE OF THE 356-KeV ROI
1250 REM
1260 REM THESE PARAMETERS SHOULD BE CHANGED AT THE SOURCE CODE LEVEL
1270 REM IF THIS PROGRAM IS TO BE USED ON A NEW SYSTEM.
1280 REM
1290 REM*****
1300 REM
1310 REM
1320 LPRINT ";; V3 AD P1 8,0,8,1900 U "
1330 LPRINT ";; 8,440 D 0,440 U 0,880 D 8,880 U 8,1319 D 0,1319 U 0,1759 D 8,1759 U"
1340 REM
1350 ADVANCE =PAPER.ADVANCE
1360 REM
1370 REM*****
1380 REM***** SET UP AND INITIALIZE PCA DISPLAY *****
1390 REM*****
1400 REM
1410 CLS
1420 REM
1430 SCREEN 2 ' HIGH RESOLUTION GRAPHICS MODE
1440 F=2 ' DISPLAY INITIALIZATION
1450 GOSUB 6430 ' GOTO POKING SUBROUTINE
1460 F=3 ' PARAMETERS UPDATE
1470 GOSUB 6430 ' GOTO POKING SUBROUTINE
1480 F=4 ' SPECTRUM UPDATE
1490 GOSUB 6430 ' GOTO POKING SUBROUTINE
1500 REM
1510 REM*****
1520 REM***** SET PRESET TIME *****
1530 REM*****
1540 REM
1550 REM PRESET TIME WILL BE 30 SECONDS.
1560 REM
1570 REM THE ACQUISITION PRESET TIME FUNCTION IS A
1580 REM 3-BYTE PACKED BINARY CODED DECIMAL.
```

PROGRESS continued

```
1590 REM
1600 PT=PRESET.TIME
1610 REM
1620 PT1=INT(P/10000)          ' BCD NUMBER FOR ONES AND TENS
1630 PT2=INT((P-PT1*10000)/100) ' BCD NUMBER FOR HUNDREDS & THOUSANDS
1640 PT3=INT(P-PT1*10000-PT2*100) ' BCD # FOR 10 THOUSANDS AND 100 1000'S
1650 REM
1660 REM
1670 REM      CONVERT DECIMAL VALUES TO HEX.
1680 REM
1690 PT1=INT(P/10)*16+INT(P-(INT(P/10)*16)*(10/16))
1700 PT2=INT(P/10)*16+INT(P-(INT(P/10)*16)*(10/16))
1710 PT3=INT(P/10)*16+INT(P-(INT(P/10)*16)*(10/16))
1720 REM
1730 BYTE1=PT3          ' ASSIGN VARIABLES TO PACK BCD NUMBERS
1740 BYTE2=PT2
1750 BYTE3=PT1
1760 REM
1770 F=141          ' ACQUISITION PRESET TIME FUNCTION NUMBER
1780 REM
1790 GOSUB 6140      ' GOTO SUBROUTINE TO REPLACE VALUE IN MEMORY
1800 REM
1810 F=25          ' PHA PRESET, ELAPSED AND REMAINING TIME DISPLAY
1820 REM      THIS MUST BE CALLED TO COMPLETE THE TIME SELECTION.
1830 REM
1840 GOSUB 6430      ' GOTO POKING SUBROUTINE
1850 REM
1860 REM*****
1870 REM***** END PRESET TIME SETTING *****
1880 REM*****
1890 F=158
1900 GOSUB 5900
1910 PRINT VALUE0,VALUE1,TOTAL
1920 IF VALUE0=1 THEN 1970
1930 F=39
1940 GOSUB 6520
1950 GOTO 1890
1960 REM
1970 REM*****
1980 REM***** SET BA ROIs *****
1990 REM*****
2000 REM
2010 REM
2020 REM
2030 REM      BA.1.ROISTART, BA.1.ROIEND, BA.2.ROISTART, AND BA.2.ROIEND
2040 REM      ARE DEFINED IN THE PARAMETERS LOOP OR IN THE DEFAULT SECTION.
2050 REM
2060 REM      THE REST OF THE ROI SETTINGS ARE DEPENDENT ON THESE TWO ROIs.
2070 REM      THEY SHOULD BRACKET THE Ba-133 81-KeV AND 356-KeV PEAKS.
2080 REM
2090 REM      THE ENERGY CALIBRATION PERFORMED WITH THESE PEAKS
2100 REM      IS USED TO CALCULATE THE CHANNEL # FOR THE REMAINING ROIs.
2110 REM
2120 REM
```

PROGRESS continued

```
2130 ROISTART=BA.1.ROISTART:ROIEND=BA.1.ROIEND      'Ba-133 80.998 KeV PEAK
2140 REM
2150 GOSUB 5620      ' GOTO THE SET ROI SUBROUTINE
2160 REM
2170 ROISTART=BA.2.ROISTART:ROIEND=BA.2.ROIEND      'Ba-133 356.005 KeV PEAK
2180 REM
2190 GOSUB 5620      ' GOTO THE SET ROI SUBROUTINE
2200 REM
2210 REM
2220 REM
2230 REM*****
2240 REM***** END SET Ba ROIs *****
2250 REM*****
2260 REM
2270 REM
2280 REM*****
2290 REM***** CALCULATE Ba CENTROIDS *****
2300 REM*****
2310 REM
2320 REM
2330 REM      GOTO.BA.1 AND GOTO.BA.2 ARE VARIABLES THAT CALCULATE
2340 REM      THE MIDPOINTS OF THE TWO Ba ROIs.
2350 REM
2360 REM      THE CURSOR MUST BE INSIDE AN ROI BEFORE THE CENTROID
2370 REM      CAN BE CALCULATED.
2380 REM
2390 GOTO.BA.1=INT(((BA.1.ROIEND-BA.1.ROISTART)/2)+BA.1.ROISTART)
2400 GOTO.BA.2=INT(((BA.2.ROIEND-BA.2.ROISTART)/2)+BA.2.ROISTART)
2410 REM
2420 REM
2430 REM      THE FOLLOWING STATEMENTS DETERMINE THE VALUES OF THE
2440 REM      2-BYTE BINARY NUMBERS THAT INDICATE THE CURRENT
2450 REM      CHANNEL NUMBER OF THE CURSOR.
2460 REM
2470 REM
2480 IF GOTO.BA.1>256 THEN GOSUB 7840
2490 IF GOTO.BA.1<256 THEN GOSUB 7880
2500 IF GOTO.BA.1=256 THEN GOSUB 7920
2510 REM
2520 REM
2530 REM
2540 REM
2550 GOSUB 6720      ' GOTO THE READ CENTROID SUBROUTINE
2560 REM
2570 BACENT1=C      ' CHANNEL NUMBER OF THE FIRST Ba CENTROID
2580 REM
2590 REM
2600 IF GOTO.BA.2<1280 THEN GOSUB 7960 'BYTE1=GOTO.BA.2 - 1024 AND BYTE2=4
2610 IF GOTO.BA.2>1280 AND GOTO.BA.2<1536 THEN GOSUB 8010 'BYTE1=GOTO.BA.2 - 1280 AND
BYTE2=5
2620 IF GOTO.BA.2>1536 THEN GOSUB 8060 'BYTE1=GOTO.BA.2 - 1536 AND BYTE2=6
2630 REM
2640 GOSUB 6720      ' GOTO THE READ CENTROID SUBROUTINE
2650 REM
```


PROGRESS continued

```
2660 BACENT2=C      ' CHANNEL NUMBER OF THE SECOND Ba CENTROID
2670 REM
2680 REM
2690 DELTAENERGY=275.017      ' 356.005 KeV - 80.998 KeV = 275.017 KeV
2700 REM
2710 DELTACHANNEL=BACENT2-BACENT1 ' DIFF IN CHANNELS BETWEEN TWO CENTROIDS
2720 REM
2730 REM
2740 REM
2750 REM
2760 REM      CALCULATE ENERGY CALIBRATION
2770 REM
2780 REM
2790 REM
2800 REM
2810 REM      ENERGY CALIBRATION EQUATION:
2820 REM
2830 REM      Y = mX + b
2840 REM
2850 REM      or
2860 REM
2870 REM      CHANNEL NUMBER = (SLOPE)(ENERGY OF PEAK) + YINTERCEPT
2880 REM
2890 REM
2900 REM      CALCULATE THE SLOPE
2910 REM
2920 SLOPE=DELTACHANNEL/DELTAENERGY
2930 REM
2940 REM      CALCULATE THE YINTERCEPT
2950 REM
2960 YINTERCEPT=BACENT2-(SLOPE*356.005)
2970 REM
2980 REM      CALCULATE THE CHANNEL NUMBERS OF ALL THE CENTROIDS.
2990 REM
3000 AMSTR=INT(SLOPE*56.784+YINTERCEPT) ' LEFT SIDE OF Am ROI
3010 AMEND=INT(SLOPE*61.084+YINTERCEPT) ' RIGHT SIDE OF Am ROI
3020 REM
3030 PWSTR=INT(SLOPE*127.447+YINTERCEPT) ' LEFT SIDE OF 1st Pu ROI
3040 PUEND=INT(SLOPE*130.887+YINTERCEPT) ' RIGHT SIDE OF 1st Pu ROI
3050 REM
3060 USTR=INT(SLOPE*205.505+YINTERCEPT) ' LEFT SIDE U ROI
3070 UEND=INT(SLOPE*209.591+YINTERCEPT) ' RIGHT SIDE OF U ROI
3080 REM
3090 PU2STR=INT(SLOPE*410.796+YINTERCEPT) ' LEFT SIDE OF 2nd Pu ROI
3100 PU2END=INT(SLOPE*415.798+YINTERCEPT) ' RIGHT SIDE OF 2nd PU ROI
3110 REM
3120 REM
3130 REM*****
3140 REM***** SET REGIONS OF INTEREST *****
3150 REM***** PARAMETERS *****
3160 REM*****
3170 REM
3180 REM
3190 REM      SET FIRST ROI ABOUT Am-241 59.5-KeV PEAK.
```

PROGRESS continued

```
3200 REM
3210 ROISTART=AMSTR:ROIEND=AMEND
3220 REM
3230 GOSUB 5620      ' GOTO THE SET ROI SUBROUTINE
3240 REM
3250 REM
3260 REM          SET SECOND ROI ABOUT Pu-239 129.3-KeV PEAK.
3270 REM
3280 ROISTART=PUSTR:ROIEND=PUEND
3290 REM
3300 GOSUB 5620      ' GOTO THE SET ROI SUBROUTINE
3310 REM
3320 REM
3330 REM          SET THIRD ROI ABOUT U-237 208-KeV PEAK.
3340 REM
3350 ROISTART=USTR:ROIEND=UEND
3360 REM
3370 GOSUB 5620      ' GOTO THE SET ROI SUBROUTINE
3380 REM
3390 REM
3400 REM          SET FOURTH ROI ABOUT Pu-239 414-KeV PEAK.
3410 REM
3420 'ROISTART=PU2STR:ROIEND=PU2END
3430 REM
3440 'GOSUB 37700      ' GOTO THE SET ROI SUBROUTINE
3450 REM
3460 REM
3470 REM*****
3480 REM***** END SET ROI PARAMETERS *****
3490 REM*****
3500 REM
3510 REM
3520 REM*****
3530 REM***** START DATA ACQUISITION *****
3540 REM*****
3550 REM
3560 REM
3570 F=6          ' ERASE SPECTRAL DATA
3580 REM
3590 GOSUB 6540      ' GOTO POKING SUBROUTINE
3600 REM
3610 REM
3620 BEEP
3630 REM
3640 F=5          ' DATA ACQUISITION START/STOP
3650 REM
3660 GOSUB 6430      ' GOTO POKING SUBROUTINE
3670 REM
3680 REM
3690 REM*****
3700 REM***** PLOTTING LOOP *****
3710 REM*****
3720 DEF SEG
3730 ADVANCE=ADVANCE+ADVANCE.INTERVAL
```

PROGRESS continued

```
3740 IF AM241=0 THEN AM241=1
3750 IF PU241=0 THEN PU241=1
3760 IF U237=0 THEN U237=1
3770 AM241ADJ=INT(LOG(AM241/PRESET.TIME)*200)
3780 PU241ADJ=INT(LOG(PU241/PRESET.TIME)*200)
3790 U237ADJ=INT(LOG(U237/PRESET.TIME)*200)
3800 IF AM241ADJ<1 THEN AM241ADJ=1
3810 IF PU241ADJ<1 THEN PU241ADJ=1
3820 IF U237ADJ<1 THEN U237ADJ=1
3830 REM*****
3840 REM      HOUSTON INSTRUMENTS MODEL #695 PLOTTER SUBROUTINE
3850 REM
3860 LPRINT ";; A P1"+STR$(ADVANCE)+",""+STR$(AM241ADJ)+"S12 A _"
3870 LPRINT ";; A P2"+STR$(ADVANCE)+",""+STR$(PU241ADJ)+"S12 P _"
3880 LPRINT ";; A P3"+STR$(ADVANCE)+",""+STR$(U237ADJ)+"S12 U _"
3890 IF TRIGGER=1 THEN LPRINT ";; A P3"+STR$(ADVANCE)+",1885"+"S12 |_"
3900 LPRINT ";; "+STR$(ADVANCE+1000)+",""+STR$(U237ADJ)
3910 REM
3920 TRIGGER=0
3930 REM
3940 REM
3950 REM
3960 REM
3970 F=3      ' PARAMETERS UPDATE FUNCTION
3980 REM
3990 GOSUB 6430      ' GOTO POKING SUBROUTINE
4000 REM
4010 INTERRUPTS$=INKEY$
4020 REM
4030 IF INTERRUPTS$=CHR$(0)+CHR$(113) THEN 3390      'GOTO MENU
4040 REM
4050 IF INTERRUPTS$=CHR$(0)+CHR$(104) THEN TRIGGER=1
4060 REM
4070 IF TRIGGER=1 THEN SOUND 4000,2
4080 REM
4090 REM      BEGIN CHECKING LOOP FOR DATA ACQUISITION FLAG ON/OFF.
4100 REM
4110 F=4      ' SPECTRUM UPDATE FUNCTION
4120 REM
4130 GOSUB 6430      ' GOTO POKING SUBROUTINE
4140 REM
4150 REM
4160 REM
4170 F=140      ' CHECK ACQUISITION FLAG FOR PRESET TIME-OUT
4180 REM
4190 GOSUB 5910      ' GOTO PEEKING SUBROUTINE
4200 REM
4210 IF VALUE0=0 THEN 3970      ' A VALUE OF "0" MEANS SYSTEM IS ACQUIRING DATA
4220 REM
4230 CLS
4240 REM
4250 REM
4260 REM*****
4270 REM***** END DATA ACQUISITION *****
```

PROGRESS continued

```
4280 REM*****
4290 REM
4300 REM
4310 REM*****
4320 REM***** MOVE INTO ROIs AND CALCULATE NET INTEGRAL *****
4330 REM*****
4340 REM
4350 REM
4360 REM      CALCAMCENT, CALCPUCENT, CALCUCENT and CALCPU2CENT ARE
4370 REM      VARIABLES USED TO ENTER EACH ROI
4380 REM      TO CALCULATE EACH NET INTEGRAL.
4390 REM
4400 REM
4410 REM
4420 REM      GOTO CHANNEL #COLCAMCENT, Am-241 PEAK 59.5 KeV
4430 REM
4440 CALCAMCENT=INT(SLOPE*59.5+YINTERCEPT)  'CALC CHANNEL FOR 59.5 KeV
4450 REM
4460 REM
4470 IF CALCAMCENT<256 THEN GOSUB 8140
4480 IF CALCAMCENT=256 THEN GOSUB 8160
4490 IF CALCAMCENT>256 THEN GOSUB 8180
4500 REM
4510 GOSUB 7680      ' GOTO NET INTEGRAL READING SUBROUTINE
4520 REM
4530 AM241=TOTALNET  ' Am-241 IS THE NET INTEGRAL VALUE FOR THAT ROI
4540 REM
4550 REM
4560 REM
4570 REM      GOTO CHANNEL #CALCPUCENT, Pu-239 PEAK 129.3 KeV
4580 REM
4590 CALCPUCENT=INT(SLOPE*129.3+YINTERCEPT)  ' CALC CHANNEL # FOR 129.3 KeV
4600 REM
4610 IF CALCPUCENT<512 THEN GOSUB 8200
4620 IF CALCPUCENT=512 THEN GOSUB 8220
4630 IF CALCPUCENT>512 THEN GOSUB 8240
4640 REM
4650 GOSUB 7680      ' GOTO NET INTEGRAL READING SUBROUTINE
4660 REM
4670 PU239=TOTALNET  ' Pu-239 IS THE NET INTEGRAL VALUE FOR THAT ROI
4680 REM
4690 REM
4700 REM
4710 REM      GOTO CHANNEL #CALCUCENT, U-237 PEAK, 208.0 KeV
4720 REM
4730 CALCUCENT=INT(SLOPE*208+YINTERCEPT)  ' CALC CHANNEL NUMBER FOR 208 KeV
4740 REM
4750 IF CALCUCENT<768 THEN GOSUB 8260
4760 IF CALCUCENT=768 THEN GOSUB 8280
4770 IF CALCUCENT>768 THEN GOSUB 8300
4780 REM
4790 GOSUB 7680      ' GOTO THE NET INTEGRAL READING SUBROUTINE
4800 REM
4810 U237=TOTALNET  ' U-237 IS THE NET INTEGRAL VALUE
```

PROGRESS continued

```

4820 REM
4830 REM
4840 REM
4850 GOTO 4990
4860 REM
4870 REM          GOTO CHANNEL #CALCPU2CENT, Pu-239 PEAK, 413.7 KeV
4880 REM
4890 CALCPU2CENT=INT(SLOPE*413.7+YINTERCEPT) ' CALC CHANNEL # FOR 413.7 KeV
4900 REM
4910 IF CALCPU2CENT<1792 THEN GOSUB 8320
4920 IF CALCPU2CENT=1792 THEN GOSUB 8340
4930 IF CALCPU2CENT>1792 THEN GOSUB 8360
4940 REM
4950 REM
4960 GOSUB 7680      ' GOTO THE NET INTEGRAL READING SUBROUTINE
4970 REM
4980 PU239NO2=TOTALNET ' PU-239NO2 IS THE NET INTEGRAL VALUE
4990 REM
5000 REM
5010 REM*****
5020 REM***** CLEAR ROI'S AND RETURN TO THE *****
5030 REM***** BEGINNING OF THE PROGRAM *****
5040 REM*****
5050 REM
5060 REM
5070 REM
5080 ROICLEARSTR=AMSTR-1:ROICLEAREND=AMEND+1 ' CLEAR Am ROI
5090 REM
5100 GOSUB 5760      ' GOTO ROI CLEAR SUBROUTINE
5110 REM
5120 REM
5130 ROICLEARSTR=PUSTR-1:ROICLEAREND=PUEND+1 ' CLEAR Pu ROI
5140 REM
5150 GOSUB 5760      ' GOTO ROI CLEAR SUBROUTINE
5160 REM
5170 REM
5180 ROICLEARSTR=USTR-1:ROICLEAREND=UEND+1 ' CLEAR U ROI
5190 REM
5200 GOSUB 5760      ' GOTO ROI CLEAR SUBROUTINE
5210 REM
5220 REM
5230 'ROICLEARSTR=PU2STR-1:ROICLEAREND=PU2END+1 ' CLEAR 2nd PU SUBROUTINE
5240 REM
5250 'GOSUB 38900      ' GOTO ROI CLEAR SUBROUTINE
5260 REM
5270 REM
5280 CLS
5290 REM
5300 REM
5310 REM
5320 GOTO 2280      ' RETURNS TO THE SET Ba SUBROUTINE
5330 REM
5340 REM
5350 REM

```

PROGRESS continued

```
5360 REM
5370 REM
5380 REM
5390 REM
5400 REM
5410 REM      END MAIN BODY OF PROGRAM
5420 REM      END MAIN BODY OF PROGRAM
5430 REM      END MAIN BODY OF PROGRAM
5440 REM
5450 REM
5460 REM
5470 REM
5480 REM
5490 REM*****
5500 REM*****
5510 REM
5520 REM      BEGIN SECTION CONTAINING
5530 REM      ALL SUBROUTINES IN PROGRAM
5540 REM
5550 REM*****
5560 REM*****
5570 REM
5580 REM*****
5590 REM***** SET REGIONS OF INTEREST *****
5600 REM***** SUBROUTINE *****
5610 REM*****
5620 DEF SEG=&HE000
5630 FOR I=ROISTART TO ROIEND
5640 ROI=PEEK(I*4 +3)
5650 IF ROI>127 THEN 5680
5660 ROI=ROI+128
5670 POKE (I*4 +3),ROI
5680 NEXT I
5690 REM*****
5700 REM***** END SET ROI SUBROUTINE *****
5710 REM*****
5720 RETURN
5730 REM*****
5740 REM***** CLEAR ALL ROIs SUBROUTINE *****
5750 REM*****
5760 REM
5770 DEF SEG=&HE000
5780 FOR ROICLEAR=ROICLEARSTR TO ROICLEARSTR+ROICLEAREND-ROICLEARSTR
5790 ROI =PEEK(ROICLEAR*4+3)
5800 IF ROI<128 THEN 5830
5810 ROI=ROI-128
5820 POKE(ROICLEAR*4+3),ROI
5830 NEXT ROICLEAR
5840 REM
5850 REM*****
5860 REM*****
5870 RETURN
5880 REM*****
5890 REM***** PEEKING SUBROUTINE *****
```

PROGRESS continued

```
5900 REM*****
5910 REM
5920 REM
5930 REM
5940 DEF SEG=&H0
5950 PCAOFFSET=PEEK(&H3C9)*256 + PEEK(&H3C8)
5960 PCASEG=PEEK(&H3CB)*256 +PEEK(&H3CA)
5970 PCAFUNN= PCAOFFSET+3
5980 DEF SEG=PCASEG
5990 DEF USR=PCAOFFSET
6000 POKE PCAFUNN,F
6010 X=USR(0)
6020 PARMOFFSET=PEEK(PCAFUNN+2)*256 +PEEK(PCAFUNN+1)
6030 PARMSEG=PEEK(PCAFUNN+4)*256 + PEEK(PCAFUNN+3)
6040 DEF SEG=PARMSEG
6050 VALUE0=PEEK(PARMOFFSET)
6060 VALUE1=(PEEK(PARMOFFSET+1))*256
6070 TOTAL=VALUE0+VALUE1
6080 RETURN
6090 REM
6100 REM*****
6110 REM***** END PEEKING SUBROUTINE *****
6120 REM*****
6130 REM
6140 REM
6150 REM*****
6160 REM***** SUBROUTINE TO REPLACE VALUE IN MEMORY *****
6170 REM***** USING THE POKE FUNCTION *****
6180 REM*****
6190 REM
6200 REM
6210 REM          SUBROUTINE TO REPLACE VALUE IN MEMORY OFFSET
6220 REM          USING THE POKE FUNCTION
6230 REM
6240 REM
6250 DEF SEG=&H0
6260 PCAOFFSET=PEEK(&H3C9)*256 + PEEK(&H3C8)
6270 PCASEG=PEEK(&H3CB)*256 +PEEK(&H3CA)
6280 PCAFUNN= PCAOFFSET+3
6290 DEF SEG=PCASEG
6300 DEF USR=PCAOFFSET
6310 POKE PCAFUNN,F
6320 X=USR(0)
6330 PARMOFFSET=PEEK(PCAFUNN+2)*256 +PEEK(PCAFUNN+1)
6340 PARMSEG=PEEK(PCAFUNN+4)*256 + PEEK(PCAFUNN+3)
6350 DEF SEG=PARMSEG
6360 VALUE0=PEEK(PARMOFFSET)
6370 VALUE1=(PEEK(PARMOFFSET+1))*256
6380 TOTAL=VALUE0+VALUE1
6390 POKE PARMOFFSET,BYTE1
6400 POKE (PARMOFFSET+1),BYTE2
6410 POKE (PARMOFFSET+2),BYTE3
6420 RETURN
6430 REM
```

PROGRESS continued

```
6440 REM*****
6450 REM***** END SUBROUTINE TO REPLACE MEMORY VALUE *****
6460 REM*****
6470 REM
6480 REM
6490 REM
6500 REM*****
6510 REM***** POKING SUBROUTINE *****
6520 REM*****
6530 REM
6540 REM
6550 DEF SEG=&H0
6560 PCAOFFSET=PEEK(&H3C9)*256 + PEEK(&H3C8)
6570 PCASEG=PEEK(&H3CB)*256 +PEEK(&H3CA)
6580 PCAFUNN= PCAOFFSET+3
6590 DEF SEG=PCASEG
6600 DEF USR=PCAOFFSET
6610 POKE PCAFUNN,F
6620 X=USR(0)
6630 RETURN
6640 REM
6650 REM*****
6660 REM***** END POKING SUBROUTINE *****
6670 REM*****
6680 REM
6690 REM
6700 REM
6710 REM
6720 REM*****
6730 REM***** READ CENTROID SUBROUTINE *****
6740 REM*****
6750 REM
6760 REM
6770 DEF SEG
6780 F=155
6790 GOSUB 6140
6800 F=34
6810 GOSUB 6430
6820 F=15
6830 GOSUB 6550
6840 REM
6850 F=169
6860 DEF SEG=&H0
6870 PCAOFFSET=PEEK(&H3C9)*256 + PEEK(&H3C8)
6880 PCASEG=PEEK(&H3CB)*256 +PEEK(&H3CA)
6890 PCAFUNN= PCAOFFSET+3
6900 DEF SEG=PCASEG
6910 DEF USR=PCAOFFSET
6920 POKE PCAFUNN,F
6930 X=USR(0)
6940 PARMOFFSET=PEEK(PCAFUNN+2)*256 +PEEK(PCAFUNN+1)
6950 PARMSEG=PEEK(PCAFUNN+4)*256 + PEEK(PCAFUNN+3)
6960 DEF SEG=PARMSEG
6970 A=PARMOFFSET
```


PROGRESS continued

```
6980 C1=PEEK(A)
6990 C1U=INT(C1/16)
7000 C1L=C1-(C1U*16)
7010 REM
7020 REM
7030 C2=PEEK(A+1)
7040 C2U=INT(C2/16)
7050 C2L=C2-(C2U*16)
7060 REM
7070 REM
7080 C3=PEEK(A+2)
7090 C3U=INT(C3/16)
7100 C3L=C3-(C3U*16)
7110 REM
7120 REM
7130 C4=PEEK(A+3)
7140 C4U=INT(C4/16)
7150 C4L=C4-(C4U*16)
7160 REM
7170 REM
7180 C5=PEEK(A+4)
7190 C5U=INT(C5/16)
7200 C5L=C5-(C5U*16)
7210 REM
7220 REM          CM=CENTROID MANTISSA
7230 REM
7240 CM=C1L+C1U*10+C2L*100+C2U*1000+C3L*10000+C3U*100000!+C4L*1000000!+C4U*1E+07+C5L*10000-
0000#
7250 REM
7260 REM          CE=CENTROID EXPONENT
7270 CE=PEEK(A+5)
7280 C=CM/(10^(9-CE))
7290 RETURN
7300 REM
7310 REM*****
7320 REM*****  END CENTROID READING SUBROUTINE *****
7330 REM*****
7340 REM
7350 REM*****
7360 REM*****  NET INTEGRAL READING SUBROUTINE *****
7370 REM*****
7380 REM
7390 F=15
7400 GOSUB 6530
7410 DEF SEG=&H0
7420 PCAOFFSET=PEEK(&H3C9)*256 + PEEK(&H3C8)
7430 PCASEG=PEEK(&H3CB)*256 +PEEK(&H3CA)
7440 PCAFUNN= PCAOFFSET+3
7450 DEF SEG=PCASEG
7460 DEF USR=PCAOFFSET
7470 POKE PCAFUNN,166
7480 X=USR(0)
7490 PARMOFFSET=PEEK(PCAFUNN+2)*256 +PEEK(PCAFUNN+1)
7500 PARMSEG=PEEK(PCAFUNN+4)*256 + PEEK(PCAFUNN+3)
7510 DEF SEG=PARMSEG
7520 NET0=PEEK(PARMOFFSET)
7530 NET1=(PEEK(PARMOFFSET+1))*256
```

PROGRESS continued

```
7540 NET2=(PEEK(PARMOFFSET+2))*65536!
7550 NET3=(PEEK(PARMOFFSET+3))*16772216#
7560 TOTALNET=NET0+NET1+NET2+NET3
7570 REM
7580 REM
7590 F=167
7600 GOSUB 5900
7610 IF VALUE0>0 THEN TOTALNET=0
7620 RETURN
7630 REM
7640 REM*****
7650 REM***** END NET INTEGRAL READING SUBROUTINE *****
7660 REM*****
7670 REM
7680 REM
7690 REM*****
7700 REM***** BRANCHING SUBROUTINE FOR READING *****
7710 REM***** NET INTEGRALS *****
7720 REM*****
7730 REM
7740 REM
7750 F=155
7760 GOSUB 6140
7770 F=34
7780 GOSUB 6430
7790 GOSUB 7340      'BRANCHES TO THE NET INTEGRAL READING SUBROUTINE
7800 REM
7810 REM
7820 REM
7830 RETURN
7840 BYTE1=GOTO.BA.1 -256
7850 BYTE2=1
7860 BYTE3= 0
7870 RETURN
7880 BYTE1=GOTO.BA.1
7890 BYTE2=0
7900 BYTE3=0
7910 RETURN
7920 BYTE1=255
7930 BYTE2=0
7940 BYTE3=0
7950 RETURN
7960 REM
7970 BYTE1=GOTO.BA.2 - 1024
7980 BYTE2=4
7990 BYTE3=0
8000 RETURN
8010 REM
8020 BYTE1=GOTO.BA.2 - 1280
8030 BYTE2=5
8040 BYTE3=0
8050 RETURN
8060 REM
8070 BYTE1=GOTO.BA.2-1536
8080 BYTE2=6
8090 BYTE3=0
```

PROGRESS continued

```
8100 RETURN
8110 REM
8120 REM          BREAK INTO BYTE-SIZE PIECES.
8130 REM
8140 BYTE1=CALCAMCENT:BYTE2=0:BYTE3=0
8150 RETURN
8160 BYTE1=0:BYTE2=1:BYTE3=0
8170 RETURN
8180 BYTE1=CALCAMCENT-256:BYTE2=1:BYTE3=0
8190 RETURN
8200 BYTE1=CALCPUCENT-256:BYTE2=1:BYTE3=0
8210 RETURN
8220 BYTE1=0:BYTE2=2:BYTE3=0
8230 RETURN
8240 BYTE1=CALCPUCENT-512:BYTE2=2:BYTE3=0
8250 RETURN
8260 BYTE1=CALCUCENT-512:BYTE2=2:BYTE3=0
8270 RETURN
8280 BYTE1=0:BYTE2=3:BYTE3=0
8290 RETURN
8300 BYTE1=CALCUCENT-768:BYTE2=3:BYTE3=0
8310 RETURN
8320 BYTE1=CALCPU2CENT-1536:BYTE2=6:BYTE3=0
8330 RETURN
8340 BYTE1=0:BYTE2=7:BYTE3=0
8350 RETURN
8360 BYTE1=CALCPU2CENT-1792:BYTE2=7:BYTE3=0
8370 RETURN
8380 REM
8390 REM          INTERRUPT LOOP AND MENU.
8400 REM
8410 CLS
8420 SCREEN 0,0,0
8430 COLOR 13:LOCATE 3,23:PRINT"P":COLOR 11:LOCATE 3,24:PRINT"lutonium"
8440 COLOR 13:LOCATE 4,23:PRINT"R":COLOR 11:LOCATE 4,24:PRINT"ecoverly"
8450 COLOR 13:LOCATE 5,23:PRINT"O":COLOR 11:LOCATE 5,24:PRINT"perations"
8460 COLOR 13:LOCATE 6,23:PRINT"G":COLOR 11:LOCATE 6,24:PRINT"amma"
8470 COLOR 13:LOCATE 7,23:PRINT"R":COLOR 11:LOCATE 7,24:PRINT"ay"
8480 COLOR 13:LOCATE 8,23:PRINT"E":COLOR 11:LOCATE 8,24:PRINT"nergy"
8490 COLOR 13:LOCATE 9,23:PRINT"S":COLOR 11:LOCATE 9,24:PRINT"pectroscopy"
8500 COLOR 13:LOCATE 10,23:PRINT"S":COLOR 11:LOCATE 10,24:PRINT"ystem"
8510 COLOR 14:LOCATE 12,23:PRINT"Nuclear Materials Process Technology Group MST-12"
8520 LOCATE 13,23:PRINT"Research, Development and Demonstration"
8530 COLOR 10:LOCATE 17,23:PRINT"(1) RESUME CURRENT PROCESS MEASUREMENT"
8540 COLOR 10:LOCATE 18,23:PRINT"(2) START NEW PROCESS MEASUREMENT"
8550 COLOR 10:LOCATE 19,23:PRINT"(3) QUIT AND RETURN TO DOS"
8560 COLOR 11:LOCATE 21,23:INPUT"ENTER OPTION:",NEXT.OPTION
8570 ON NEXT.OPTION GOTO 1380,330,8590
8580 IF NEXT.OPTION>3 OR NEXT.OPTION<1 THEN 8620
8590 CLS
8600 LOCATE 10,23:COLOR 11:PRINT"TYPE SYSTEM TO RETURN TO DOS"
8610 END
8620 SOUND 500,5:LOCATE 22,23:PRINT"INCORRECT OPTION. PLEASE RE-ENTER RESPONSE"
8630 GOTO 8530
```